

Application note 012b TVM

This application note demonstrates C47/R47 TVM (Time Value of Money) functionality through 23 worked examples.

Contents

Learning outcomes:	1
Section 2: 11 problems from a worked out financial calculator tutorial.	1
Section 3: 12 numerical test cases from the calculator community as downloadable RPN programs.....	1
Introduction	2
List of worked out application note examples:	2
Section 1: Reference to HP TVM calculators:	3
Section 2: Worked out tutorial Drs. Baker & Lin	4
Section 3: From Swissmicros Forum, dm319's problem set.....	9

Learning outcomes:

Section 2: 11 problems from a worked out financial calculator tutorial.

The learning outcomes of Section 2 focus on TVM fundamentals: understanding how the five core variables interact, setting payment and compounding periods correctly, and recognizing when to use Begin versus End mode.

Examples 3A/3B/3C demonstrate that different problem approaches yield the same answer.

Example 5 teaches period conversion (annual to bi-weekly).

Examples 6 and 7 show how adding constraints changes the calculation.

Example 11 covers nominal to effective interest rate conversions.

Section 3: 12 numerical test cases from the calculator community as downloadable RPN programs.

The learning outcomes of Section 3 shift to understanding the TVM functions by programming same: structuring complete TVM programs, organizing multi-step calculations, and controlling solver behaviour.

Problems demonstrate straightforward setup and solving and progress to teach program sequencing when multiple TVM operations are required. Problems 5/5b and 6/6b specifically teach solver bias control, showing how to obtain different valid solutions by setting initial guesses.

All programs are available at 47calc.com, on the appnote section for practice and adaptation.

Introduction

The C47 TVM implementation follows HP financial calculator conventions but requires explicit setup of payment and compounding periods. The five core TVM registers are n (number of periods), i/a (annual interest rate as percentage), PV (present value), PMT (payment per period), and FV (future value). Special settings control pp/a (payment periods per year), cp/a (compounding periods per year), and Begin/End mode (when interest applies). Functions EFF/a and $DEFF/a$ convert between nominal and effective annual rates.

List of worked out application note examples:

Section 2 demonstrates typical financial calculations using problems from the Baker & Lin tutorial:

- Example 1: Calculate FV given initial investment and time horizon
- Example 2: Calculate PV of a future amount
- Example 3: Compare lottery options (lump sum vs annuity stream)
- Example 3A: Same comparison using BEGIN mode
- Example 3B/C: Alternative approach comparing future values instead of present values
- Example 4: Calculate required annual savings to reach investment goal
- Example 5: Calculate bi-weekly savings (26 periods per year)
- Example 6: Determine required rate of return given savings capacity
- Example 7: Required return with existing capital and ongoing contributions
- Example 11: Convert nominal to effective interest rates for quarterly and daily compounding

Section 3 tackles numerical edge cases from dm319 (Duncan's) calculator test suite, implemented as RPN:

- Problem 1: Monthly mortgage payment over 38 years
- Problem 1b: Reverse calculation verifying interest rate
- Problem 2: Payment differential from rate change, then PV calculation
- Problem 3: Kahan's extreme test ($60 \times 60 \times 24 \times 365$ periods)
- Problem 4: Zero interest rate round-trip (exposes floating point precision limits)
- Problem 5: Multiple solution test (default solver finds first root at 14.44%)
- Problem 5b: Biased solver finds second root at 53.17%
- Problem 6: Multiple solution test (default finds -36.89%)
- Problem 6b: Biased solver finds second root at 58.46%
- Problem 7: Near-zero FV with fractional payments
- Problem 8: Small interest rate precision test
- Problem 9: Period calculation in END mode
- Problem 10: Period calculation in BEGIN mode
- Problem 11: Sequential FV then i/a calculation
- Problem 12: Sequential i/a then FV calculation

Program 'Pall' executes the complete test suite with timing, and subroutine 'TT' logs results to text files. Execution times on USB-powered DM42 hardware range from 0.4 to 4.3 seconds per problem.

Section 1: Reference to HP TVM calculators:

This AppNote is to demonstrate the TVM menu, using an available web tutorial.

Note the C47 TVM key terminology is slightly more detailed but essentially the same as on the HP10bII and other financial HP calculators:



Photo credit of HP10bII+, credit hp.com

PMT = Payment =						-9 410,264
Begin ☐	End ☒	pp/a ₁₂	cp/a ₁₂	CLTVM	CASHFL	
RCL n	RCL I/a	RCL PV	RCL PMT	RCL FV	EFF→I/a	
n ₃₀₀	I/a _{12.3}	PV ₈₇₅₀₀₀	P _{-9410.26}	FV ₀	I→EFF/a	

C47 TVM Menu with typical values populated, version 00.109.02.06b5

5 Standard TVM keys:

- n** ⇒ This key refers to the number of (payment) periods
- I/a** ⇒ This key refers to the interest rate (10% would be 10 not 0.10). a.k.a. discount rate or rate of return.
- PV** ⇒ This key refers to the Present Value
- PMT** ⇒ This key refers to the Annuity Payment
- FV** ⇒ This key refers to the Future Value

Special options:

- pp/a** Payment periods per year, generally 1 or 12.
- cp/a** Compounding periods per year, generally 1 or 12.
- Begin ☐ End ☒** Interest on the beginning of the period, or end of the period.

Section 2: Worked out tutorial Drs. Baker & Lin

From the great tutorial by Dr. Kevin Bracker; Dr. Fang Lin; and jpursley at [Pressbooks](#) or [Wayback Machine](#):

All references below coloured in yellow are directly taken from [linked](#) document, under the appropriate Creative Commons license NC4.0 by in this referenced document.

Example One:

6 n 7 I/a 10 000 PV 0 PMT Here, on C47 you have to specify the details of compounding. Do 1 pp/a and cp/a will be defaulted to the same. Ensure End is set: Begin End . FV is -15 007.30	You are investing \$10,000 today and want to know how much you will have after 6 years if you earn a 7% rate of return over the 6-year time frame. Since you are starting with \$10,000, that is your present value. You have 6 years, so the number of time periods is 6. The 7% rate of return means you have a 7% interest rate. In this example we are not using an annuity, so we are going to set the Annuity Payment to zero. 6 N 7 I/Y 10,000 PV 0 PMT FV -15,007.30
--	---

Example Two:

5 n 9 I/a 0 PMT 6 000 FV Here, on C47 you have to specify the details of compounding. Do 1 pp/a and cp/a will be defaulted to the same. Ensure End is set: Begin End . PV is -3 899.59	You are going to receive \$6000 in 5 years. Assuming a 9% discount rate, what is this worth to you today? 5 N 9 I/Y 0 PMT 6000 FV PV -3899.59
--	--

Note: light grey represents settings the same as the previous problem, i.e. no need to re-enter those.

Example Three:

25 n 6 I/a 400 000 PMT 0 FV Here, on C47 you have to specify the details of compounding. Do 1 pp/a and cp/a will be defaulted to the same. Ensure End is set: ☐ Begin ☒ End ☒ PV is -5 113 342.46	Assume you have just won a \$10 Million Lottery Jackpot. However, instead of paying you the \$10 Million up front, you have the choice of receiving \$5 Million today or \$400,000 per year at the end of each year for the next 25 years. Assuming a 6% discount rate, which would you prefer? In order to answer this, you need to find the PV of the \$400,000 per year for 25 years. This is done as follows: 25 N 6 I/Y 400,000 PMT 0 FV PV 5,113,342.46 (Note that we dropped the negative sign) To get back to the original question, receiving the \$10M in total over 25 years is clearly better than receiving \$5M now!
---	--

Note: light grey represents settings the same as the previous problem, i.e. no need to re-enter those.

Example Three A:

All the same as Ex 3, set to BEGIN: ☒ Begin ☐ End ☒ PV is -5 420 143.01	To be more realistic, as the first 400k will be paid immediately, set to Begin, and see the 'deal' being worth even more making it the right decision to take the deal. PV 5,420,143.01 (Note that we dropped the negative sign)
--	--

Example Three B & C: calculating using future end values

C47 Solution B: 25 n 6 I/a 5 000 000 PV 0 PMT 1 pp/a ☐ Begin FV is -21 459 353.60 C47 Solution C: 25 n 6 I/a 0 PV 400 000 PMT 1 pp/a ☐ Begin FV is -23 262 553.09	Although the original solution THREE A (in yellow above) is the most elegant, a more "hand-waiving" intuitive way (as opposed to the quickest way) to illustrate the compounding comparison better, is equally easily done on the C47: We simply compare the tangible future values of both deals instead of the 'financial' way of evaluating the current worth of a future investment: Clearly the same answer, that \$23.3M is more convincing than \$21.5M: B: We first check what \$5M would do for us in 25 years: 25 N 6 I/Y 5 000 000 PV 0 PMT FV \$21,459,353.60 (dropped the negative sign) C: Then what the "deal" would do for us in the same 25 years: 25 N 6 I/Y 0 PV 400 000 PMT FV \$23,262,553.09 (dropped the negative sign)
--	---

Example Four:

20 n 10 I/a 0 PV 1 000 000 FV Here, on C47 you have to specify the details of compounding. Do 1 pp/a and cp/a will be defaulted to the same. Ensure End is set: Begin End PMT is -17 459.62	You want to become a millionaire and plan to do so through a savings/investment plan. Assuming you want to reach your goal in 20 years and anticipate earning a 10% rate of return, how much must you save at the end of each year in order to reach your goal? 20 N 10 I/Y 0 PV 1,000,000 FV PMT \$17,459.62 This means you will need to save \$17,459.62 per year in order to achieve your goal.
--	---

Note: light grey represents settings the same as the previous problem, i.e. no need to re-enter those.

Example Five:

C47 has the setting Payment Periods Per Year, so type 26 pp/a and it is assumed the Compounding Periods Per Year cp/a is the same. 30 ENTER 26 X n 10 I/a (still the same as Ex 4) 0 PV (still the same as Ex 4) 1 000 000 FV (still the same as Ex 4) Ensure End is set: Begin End (still the same as Ex 4). PMT is -202.75	Now assume that you want to accumulate \$1 million in 30 years, but instead of saving each year, you are going to save every two weeks (we will earn a 10% annual rate of return). There are 26 2-week periods in each year, so now you have to adjust your calculator to work with 26 periods per year. You can do this as follows: 26 SHIFT P/YR 780 N 10 I/Y 0 PV 1,000,000 FV PMT \$202.75
---	---

Note: light grey represents settings the same as the previous problem, i.e. no need to re-enter those.

Example Six:

C47 has the setting Payment Periods Per Year, so type 1 pp/a and it is assumed the Compounding Periods Per Year cp/a is the same. (still the same as Ex 4, 5) 35 n -3 000 PMT 0 PV (still the same form Ex 4, 5) 1 000 000 FV (still the same form Ex 4, 5) Ensure End is set: Begin ☐ End ☒ (still the same as Ex 4, 5). I/a is 10.89 %	Let's keep working with the goal of becoming a millionaire. However, instead of calculating how much you must save, we'll assume you can save \$3000 per year and want to find the rate of return you will need to earn to reach your goal. This time we will give ourselves 35 years of saving \$3000 per year. 35 N 0 PV -3000 PMT 1,000,000 FV I/Y 10.89%
---	--

Note: light grey represents settings the same as the previous problem, i.e. no need to re-enter those.

Example Seven:

C47 has the setting Payment Periods Per Year, so type 1 pp/a and it is assumed the Compounding Periods Per Year cp/a is the same. (still the same as Ex 4, 5, 6) 35 n (still the same as Ex 6) -3 000 PMT (still the same as Ex 6) -40 000 PV 1 000 000 FV (still the same as Ex 4, 5, 6) Ensure End is set: Begin ☐ End ☒ (still the same as Ex 4, 5, 6). I/a is 7.63 %	Here is one last variation on our millionaire example. This time, instead of starting with nothing, let's assume that we already have \$40,000 and plan to save an additional \$3000 per year over the next 35 years. Now, what rate of return must we earn in order to accumulate \$1,000,000 at the end of the 35th year? 35 N -40,000 PV -3000 PMT 1,000,000 FV I/Y 7.63%
---	---

Note: light grey represents settings the same as the previous problem, i.e. no need to re-enter those.

Examples Eight through Ten are “Cash Flow Problems” cannot yet be done by C47:

Example Eleven:

C47 does have EFF% from the latest version 00.109.02.00.

7.8 I/a to enter NOM% or Nominal interest value 4 cp/a for 4 compounding periods per year. EFF/a is 8.0311 % (Or press 4 pp/a for 4 Payment Periods Per Year, which does not matter, but cp/a will always follow.)	4 SHIFT P/YR (this sets your periods per year to 4 for quarterly compounding. Be careful here as this means all your time value of money calculations will use 4 periods per year until you change your P/YR again...just like if you changed the P/YR for a 5-key problem.) 7.8 SHIFT NOM% (this enters the 7.8% nominal rate) SHIFT EFF% (this solves for the effective annual rate to generate your final answer of 8.03%)
7.6 I/a to enter NOM% or Nominal interest value 365 cp/a for 365 Periods Per Year. EFF/a is 7.8954%	For the 7.6% compounded daily it is: 365 SHIFT P/YR 7.6 SHIFT NOM% SHIFT EFF% This will give you your answer of 7.90%

Section 3: From Swissmicros Forum, dm319's problem set

From the post of DM319 <https://forum.swissmicros.com/viewtopic.php?f=2&t=3987>, relating to comparative tests for TVM on multiple calculators. Here, the problem solutions are handled for C47 interest only as part of the application note for illustrating the compilation of RPN TVN programs on C47. Do follow the linked reference for updated results on all tested calculators:

#	Ref	N	I%YR	PV	PMT	FV	P/YR	Mode
1	DM	38 x 12	5.25%	270'000	?	0	12	end
1b	DM	38 x 12	?	270'000	-14'584/12	0	12	end
2	SlideRule	360	15% → 12%	100'000	?-?	0	12	end

<https://www.hpmuseum.org/forum/thread-20707.html>. Calculate PMT, given $n = 360$, $I\%YR = 15\%$, $PV = 100000$, $FV = 0$, then calculate PMT again with $I\%YR = 12\%$. Subtract the two results, and put that back into PMT, then change $n = 36$ and $I\%YR = 15\%$ again, and calculate PV.

3	Kahan 1983	60x60x24x365	10%	0	-0.01	?	=N	end
---	------------	--------------	-----	---	-------	---	----	-----

<https://www.hpmuseum.org/forum/thread-1012.html>

4	DM	480	0 → ?	100'000	? → PMT	0	12	end
---	----	-----	-------	---------	---------	---	----	-----

Calculate PMT first given $I\%YR = 0$, then re-input this back into PMT and calculate $I\%YR$. On the HP-12c this is best done by pressing $x \leftrightarrow y$ twice before putting back into PMT. **On C47 this is not required as the last solved value remains in the I/a register.**

5	Dieter	10	?	50	-30	400	1	end
6	Dieter	10	?	50	-30	80	1	end

<https://www.hpmuseum.org/cgi-sys/cgiwrap/hpmuseum/archv021.cgi?read=234439>. Note that both these problems have more than one solution.

7	A Chan	10	?	-100	10	1e-10	12	end
---	--------	----	---	------	----	-------	----	-----

<https://www.hpmuseum.org/forum/thread-18359-post-161549.html#pid161549>

8	Miguel	32	?	-999'999	0	1e6	1	end
---	--------	----	---	----------	---	-----	---	-----

<https://www.hpmuseum.org/cgi-sys/cgiwrap/hpmuseum/archv017.cgi?read=120592>

 | 9 | DM | ? | 25 | 100000 | -2083.333334 | 0 | 12 | end |
 | 10 | DM | ? | 25 | 100000 | -2040.816327 | 0 | 12 | begin |

 | 11 | robve | 60x24x365 | 1/6% → ? | 0 | -0.01 | ? → FV | =N | end |
 | 12 | robve | 40 | ? → I%YR | 900 | -400 | -1000 → ? | 1 | begin |

<https://www.hpmuseum.org/forum/thread-16565-page-2.html> Calculate for FV first, re-input back to FV and compute I%YR. 12: Calculate for i first, re-input back into i and compute FV.

- Table copied from [quoted link](#), modified typographically, 2024-09-30.
- Added text in brown.

Result of Pall program: stored file:

	Problem 1	Problem 1b	Problem 2
1-5	-1368.14835535057757212882816319868	4.373218372310086835354687826497751	6803.092161985597113625555764464724
	Problem 3	Problem 4	
1	331667.0066907768917803419084360507	ERR	
2-5	331667.0066907768917803419084360507	- 3.7593512228895892619999999999999E-19	
	Problem 5	Problem 5A	Problem 5B
1	ERR	N/A	53.17221326838472431046024137437116
2	14.43587132807995697420470129887705	14.43587132807995697420470129887705	53.17221326838472431046024137437116
3-5	14.43587132807995697420470129887705	I'm not sure how and why this is done	53.17221326838472431046024137437116
	Problem 6	Problem 6A (I called this B at the time)	Problem 6B
1	58.46195526622098709870320744645958	58.46195526622098709870320744645958	ERR
2	-36.89336987417774420723991401135426	58.46195526622098709870320744645958	-36.89336987417774420723991401135426
3-5	-36.89336987417774420723991401135426	58.46195526622098709870320744645958	I'm not sure how and why this is done
	Problem 7	Problem 8	
1	2.181818181815806921160850718651876e-10	3.125001611329216004244745444268473e-6	
2-5	2.18181818181580062818816748324244E-10	3.125001611329216004244745444268481e-6	
	Problem 9	Problem 10	
1-2	1060.303390000510007127328529884813	1076.319544367706646435106131044217	
3-4	1060.303390000510007127328529884813	1076.319544367706646435106131044440	
5	1060.303390000510007127328529884813	1076.31954436770664643510613104444	
	Problem 11	Problem 12	
1	N/A	N/A	

[illegible]

- | | | | |
|----|---------------|-----------------|------------|
| 1. | C47 1531f1cb2 | 00.109.02.00xx | 2024-05-23 |
| 2. | C47 eadc4507f | 00.109.02.00xx | 2024-06-02 |
| 3. | C47 dec8c3e | 00.109.02.06b5 | 2024-09-30 |
| 4. | C47 256756e | 00.109.02.07b5 | 2025-04-22 |
| 5. | R47 b3002aa | 00.109.03.00b2. | 2026-02-05 |

These problems were solved using RPN programs, illustrating the use thereof on C47: All programs can be downloaded from <https://47calc.com>, directly [linked here](#):

Dm319 Problem 1: <pre> 0001: LBL 'P1' 0002: 38 0003: 12 0004: * 0005: STO 'NPPER' 0006: 5.25 0007: STO 'I%/a' 0008: 270 000 0009: STO 'PV' 0010: 0 0011: STO 'PMT' 0012: 0 0013: STO 'FV' 0014: 12 0015: STO 'PPER/a' 0016: STO 'CPER/a' 0017: ENDP 0018: PMT 0019: END </pre> Result in 0.6 sec, USB powered C47 on DM42: PMT = -1368.14835535057757212882816319868	Dm319 Problem 1b: <pre> 0001: LBL 'P1b' 0002: 38 0003: 12 0004: * 0005: STO 'NPPER' 0006: 0 0007: STO 'I%/a' 0008: 270 000 0009: STO 'PV' 0010: -14 584 0011: 12 0012: / 0013: STO 'PMT' 0014: 0 0015: STO 'FV' 0016: 12 0017: STO 'PPER/a' 0018: STO 'CPER/a' 0019: ENDP 0020: I/a 0021: END </pre> Result in 0.5 sec, USB powered C47 on DM42: I/a = 4.373218372310086835354687826497751 %
Dm319 Problem 2: <pre> 0001: LBL 'P2' 0002: 360 0003: STO 'NPPER' 0004: 15 0005: STO 'I%/a' 0006: 100 000 0007: STO 'PV' 0008: 0 0009: STO 'FV' 0010: 12 0011: STO 'PPER/a' 0012: STO 'CPER/a' 0013: ENDP 0014: 0 0015: STO 'PMT' 0016: PMT 0017: 12 0018: STO 'I%/a' 0019: DROPX 0020: PMT 0021: - 0022: STO 'PMT' 0023: 36 0024: STO 'NPPER' 0025: 15 0026: STO 'I%/a' 0027: PV 0028: END </pre> Result in 1.2 sec, USB powered C47 on DM42: PV = 6803.092161985597113625555764464724	Dm319 Problem 3: <pre> 0001: LBL 'P3' 0002: 60 0003: X^2 0004: 24 0005: * 0006: 365 0007: * 0008: STO 'NPPER' 0009: STO 'PPER/a' 0010: STO 'CPER/a' 0011: 10 0012: STO 'I%/a' 0013: 0 0014: STO 'PV' 0015: -0.01 0016: STO 'PMT' 0017: ENDP 0018: FV 0019: END </pre> Result in 1.0 sec, USB powered C47 on DM42: FV = 331667.0066907768917803419084360507

Dm319 Problem 4:

```
0001: LBL 'P4'  
0002: 480  
0003: STO 'NPPER'  
0004: 0  
0005: STO 'I%/a'  
0006: 100 000  
0007: STO 'PV'  
0008: 0  
0009: STO 'PMT'  
0010: STO 'FV'  
0011: 12  
0012: STO 'PPER/a'  
0013: STO 'CPER/a'  
0014: ENDP  
0015: PMT  
0016: I/a  
0017: END
```

Result in 1.2 sec, USB powered C47 on DM42:

I/a = -3.75935122288958926199999999999999E-19 %

Dm319 Problem 5: First solution**Default result, without biasing the solver.**

```

0001: LBL 'P5'
0002: 10
0003: STO 'NPPER'
0004: 0
0005: STO 'I%/a'
0006: 50
0007: STO 'PV'
0008: -30
0009: STO 'PMT'
0010: 400
0011: STO 'FV'
0012: 1
0013: STO 'PPER/a'
0014: STO 'CPER/a'
0015: ENDP
0016: I/a
0017: END

```

Result in 0.5 sec, USB powered C47 on DM42:
I/a = 14.43587132807995697420470129887705 %

Dm319 Problem 5: Second solution

The key to the second solution is to point the solver to closer to the expected second result, i.e. store 50 in I%/a to make it search starting at 50. See **brown:**

```

0001: LBL 'P5b'
0002: 10
0003: STO 'NPPER'
0004: 0
0005: STO 'I%/a'
0006: 50
0007: STO 'PV'
0008: -30
0009: STO 'PMT'
0010: 400
0011: STO 'FV'
0012: 1
0013: STO 'PPER/a'
0014: STO 'CPER/a'
0015: ENDP
0016: 50
0017: STO 'I%/a'
0018: DROPX
0019: I/a
0020: END

```

Result in 0.5 sec, USB powered C47 on DM42:
I/a = 53.17221326838472431046024137437116 %

Dm319 Problem 6: First solution**Default result, without biasing the solver.**

```

0001: LBL 'P6'
0002: 10
0003: STO 'NPPER'
0004: 0
0005: STO 'I%/a'
0006: 50
0007: STO 'PV'
0008: -30
0009: STO 'PMT'
0010: 80
0011: STO 'FV'
0012: 1
0013: STO 'PPER/a'
0014: STO 'CPER/a'
0015: ENDP
0016: I/a
0017: END

```

Result in 0.4 sec, USB powered C47 on DM42:
I/a = -36.89336987417774420723991401135426 %

Dm319 Problem 6: Second solution

The key to the second solution is to point the solver to closer to the expected second result, i.e. store 50 in I%/a to make it search starting at 50. See **brown:**

```

0001: LBL 'P6b'
0002: 10
0003: STO 'NPPER'
0004: 0
0005: STO 'I%/a'
0006: 50
0007: STO 'PV'
0008: -30
0009: STO 'PMT'
0010: 80
0011: STO 'FV'
0012: 1
0013: STO 'PPER/a'
0014: STO 'CPER/a'
0015: ENDP
0016: 50
0017: STO 'I%/a'
0018: DROPX
0019: I/a
0020: END

```

Result in 0.7 sec, USB powered C47 on DM42:
I/a = 58.46195526622098709870320744645958 %

Dm319 Problem 7:

```

0001: LBL 'P7'
0002: 10
0003: STO 'NPPER'
0004: 0
0005: STO 'I%/a'
0006: -100
0007: STO 'PV'
0008: 10
0009: STO 'PMT'
0010: 1.*10^-10
0011: STO 'FV'
0012: 12
0013: STO 'PPER/a'
0014: STO 'CPER/a'
0015: ENDP
0016: I/a
0017: END

```

Result in 0.7 sec, USB powered C47 on DM42:
I/a = 2.18181818181580062818816748324244E-10 %

Dm319 Problem 8:

```

0001: LBL 'P8'
0002: 32
0003: STO 'NPPER'
0004: 0
0005: STO 'I%/a'
0006: -999 999
0007: STO 'PV'
0008: 0
0009: STO 'PMT'
0010: 1.*10^6
0011: STO 'FV'
0012: 1
0013: STO 'PPER/a'
0014: STO 'CPER/a'
0015: ENDP
0016: I/a
0017: END

```

Result in 0.6 sec, USB powered C47 on DM42:
I/a = 0.000003125001611329216004244745444268481 %

Dm319 Problem 9:

```

0001: LBL 'P9'
0002: 0
0003: STO 'NPPER'
0004: 25
0005: STO 'I%/a'
0006: 100 000
0007: STO 'PV'
0008: -2 083.333 334
0009: STO 'PMT'
0010: 0
0011: STO 'FV'
0012: 12
0013: STO 'PPER/a'
0014: STO 'CPER/a'
0015: ENDP
0016: n
0017: END

```

Result in 4.3 sec, USB powered C47 on DM42:
n = 1060.303390000510007127328529884813

Dm319 Problem 10:

```

0001: LBL 'P10'
0002: 0
0003: STO 'NPPER'
0004: 25
0005: STO 'I%/a'
0006: 100 000
0007: STO 'PV'
0008: -2 040.816 327
0009: STO 'PMT'
0010: 0
0011: STO 'FV'
0012: 12
0013: STO 'PPER/a'
0014: STO 'CPER/a'
0015: BeginP
0016: n
0017: END

```

Result in 3.0 sec, USB powered C47 on DM42:
n = 1076.31954436770664643510613104444

Dm319 Problem 11:

```

0001: LBL 'P11'
0002: REM 'Problem 11, first calculate FV then I/a'
0003: 60
0004: 24
0005: 365
0006: *
0007: *
0008: STO 'NPPER'
0009: STO 'PPER/a'
0010: STO 'CPER/a'
0011: 1
0012: 6
0013: /
0014: STO 'I%/a'
0015: 0
0016: STO 'PV'
0017: -0.01
0018: STO 'PMT'
0019: ENDP
0020: FV
0021: I/a
0022: END

```

[illegible]

Dm319 Problem 12:

```
0001: LBL 'P12'
0002: REM 'Problem 12 first do I/a then FV'
0003: 40
0004: STO 'NPPER'
0005: 900
0006: STO 'PV'
0007: -400
0008: STO 'PMT'
0009: -1 000
0010: STO 'FV'
0011: 1
0012: STO 'PPER/a'
0013: STO 'CPER/a'
0014: BeginP
0015: I/a
0016: FV
0017: END
```

Result in 1.0 sec, USB powered C47 on DM42:
n = -999.9999999999999999999999999999

Sequencing program 'Pall':

Run all problem files and reporting results and time durations of processing:

```

0001: LBL 'Pall'
0002: 0
0003: STO 01
0004: XEQ 'P1'
0005: 'TVM P1:'
0006: XEQ 'TT'
0007: XEQ 'P1b'
0008: 'TVM P1b:'
0009: XEQ 'TT'
0010: XEQ 'P2'
0011: 'TVM P2:'
0012: XEQ 'TT'
0013: XEQ 'P3'
0014: 'TVM P3:'
0015: XEQ 'TT'
0016: XEQ 'P4'
0017: 'TVM P4:'
0018: XEQ 'TT'
0019: XEQ 'P5'
0020: 'TVM P5:'
0021: XEQ 'TT'
0022: XEQ 'P5b'
0023: 'TVM P5B(50):'
0024: XEQ 'TT'
0025: XEQ 'P6'
0026: 'TVM P6:'
0027: XEQ 'TT'
0028: XEQ 'P6b'
0029: 'TVM P6B(50):'
0030: XEQ 'TT'
0031: XEQ 'P7'
0032: 'TVM P7:'
0033: XEQ 'TT'
0034: XEQ 'P8'
0035: 'TVM P8:'
0036: XEQ 'TT'
0037: XEQ 'P9'
0038: 'TVM P9:'
0039: XEQ 'TT'
0040: XEQ 'P10'
0041: 'TVM P10:'
0042: XEQ 'TT'
0043: XEQ 'P11'
0044: 'TVM P11:'
0045: XEQ 'TT'
0046: XEQ 'P12'
0047: 'TVM P12:'
0048: XEQ 'TT'
0049: END

```

Subroutine 'TT':

Print results to text files:

```

0001: LBL 'TT'
0002: LASTT
0003: STO 00
0004: DROPX
0005: Printxy
0006: RCL 00
0007: RCL 01
0008: -
0009: 10.0
0010: /
0011: ' Sec'
0012: SIG 02
0013: +
0014: Printx
0015: RCL 00
0016: STO 01
0017: END

```

Original 012: 2024-06-01, on C47 version 00.109.02.00
Updated 012a: **2025-04-22**, on C47 version 00.109.00.07b5
Updated 012b: **2026-02-05**, on C47/R47 version 00.109.03.00

Author Jaco Mostert; Proofread by Duncan Murray

Copyright (C) 2026 The C47 Development Team. Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.3 or any later version published by the Free Software Foundation; with no Invariant Sections, no Front-Cover Texts, and no Back-Cover Texts. A copy of the license can be downloaded from gnu.org.

Quoted yellow portions copyright to CC NC4.0 license as stated in referenced document.